
Welcome to Elixir!
    

Elixir is a dynamic, functional programming language created by José Valim. It is great for building scalable and maintainable applications. It puts the FUN into FUNctional programming! This guide will walk you through its basics in less than 30 minutes.
This guide was written by Peter Ullrich. Please submit any comments on GitHub. If you want to support me, please consider buying my courses (one and two) or my book 💜.

  
    
  
  Install Elixir


Let's first install Elixir and Erlang. You can find instructions in the official installation guide.
However, a fast way to install the language on macOS and linux is to use the asdf version manager:
# Install the elixir and erlang plugins
asdf plugin add elixir
asdf plugin add erlang

# Now install erlang
asdf install erlang latest
asdf global erlang latest

# And Elixir
asdf install elixir latest
asdf global elixir latest

# You can check that Elixir is installed with:
elixir -v


  
    
  
  Run the Code


If you want to run the following code yourself, there are three options:
	Recommended: Install Livebook (see the link for instructions) and simply click the Run in Livebook button above each file to open it in an interactive notebook. The files will open on your local computer.
	Start an Elixir REPL with iex and copy and paste the code into the command line.
	Copy and paste the code into an Elixir script file, like script.exs, and run mix run script.exs.


  
    
  
  Let's begin!


Tip
You can navigate through the pages with the left and right arrow

Let's begin: Learn how to write 'Hello World'


  

    
Hello World!
    

Here's how you write a Hello World script in Elixir.
IO.inspect("Hello World!")
"Hello World!"

  
    
  
  Variable Assignment


Assigning a value to a variable is super simple in Elixir. You just write my_var = and whatever is on the right side of the = sign gets assigned to the variable. Here is how to assign all the different basic types to variables:
name = "Peter" # String
age = 32 # Integer
age_hex = 0x20 # Integer in Hex notation
height = 190.47 # Float
height_sci = 1.9047e2 # Float in scientific notation
adult? = true # Boolean
status = :active # Atom
address = nil # The 'None/null/Nil' value

IO.inspect([name, age, age_hex, height, height_sci, adult?, status, address])
["Peter", 32, 32, 190.47, 190.47, true, :active, nil]
Variables are immutable but can be reassigned. This means that the memory holding the old value will not be overwritten; instead, new memory is allocated to hold the new value while the old value remains until garbage collected. Even if you modify a variable, it is not updated in place but copied to a new chunk of memory. This might seem wasteful, but it prevents different processes from modifying the same variable.
age = 32
age = 21
IO.inspect(age)
21
You can ignore assignments by prefixing the variable name with an underscore:
_ = 30
_result = 30 + 32

  
    
  
  String Interpolation and Concatenation


You can interpolate and concatenate strings like this:
# Interpolate a string with values
name = "Peter"
age = 32
IO.inspect("My name is #{name} and I am #{age} years old")

# Concatenate two strings
IO.inspect("But my real name is " <> "Batman 🦇")
"My name is Peter and I am 32 years old"
"But my real name is Batman 🦇"


  

    
Lists and Tuples
    

Elixir has two collection types: (Linked) Lists and Tuples.

  
    
  
  Lists


A List is defined with square brackets and can contain elements of different data types:
details = ["Peter", 32, 190.47, true, :active]
Lists are not stored contiguously in memory because they are implemented as linked lists. Each element is stored in memory with its value and a pointer to the next element. This structure makes traversing, updating, deleting, and appending values slow, but prepending new values fast. For more information, see the Big-O Time Complexities section.
You can access elements in various ways:
details = ["Peter", 32, 190.47, true, :active]

# Get the 2nd element, because lists are 0-based.
Enum.at(details, 1) # => 32

# Get the last element with a negative index
Enum.at(details, -1) # => :active

# Or use the `List` helper functions
List.first(details) # => "Peter"
List.last(details) # => :active

# Update an element at a specific index
List.replace_at(details, 1, 30)
# => ["Peter", 30, 190.47, true, :active]

# Prepend a value to the list
[:adult | details]
# => [:adult, "Peter", 32, 190.47, true, :active]

# Concatenate two lists
details ++ [nil, 123]
# => ["Peter", 32, 190.47, true, :active, nil, 123]

# Delete the first occurrence of an element
List.delete(details, 32)
# => ["Peter", 190.47, true, :active]

# Delete an element at a specific index
List.delete_at(details, -1)
# => ["Peter", 32, 190.47, true]

  
    
  
  Tuples


A Tuple is stored contiguously in memory. That makes accessing an element by its index or getting the tuple size a fast operation.
hobbies = {"Sport", "Coffee", "Writing"}

# Get the first element
elem(hobbies, 0) # => "Sport"

# Update an element
put_elem(hobbies, 2, "Reading")
# => {"Sport", "Coffee", "Reading"}

# Append an element
Tuple.append(hobbies, "Coding")
# => {"Sport", "Coffee", "Writing", "Coding"}

# Delete an element at a specific index
Tuple.delete_at(hobbies, 1)
# => {"Sport", "Writing"}

# Get the Tuple size
tuple_size(hobbies) # => 3
If you look closely at the output above, you'll see that although we update and append elements to our tuple, the original hobbies variable is not modified. In Elixir, variables are immutable and every function call => a copy of the input variable instead of a modified version of the original variable.


  

    
Maps
    

Elixir has two "associative data structures": Maps and Keyword Lists. You might know them as dictionary, map, hash table, key-value store, etc. Let's first look at Maps.
A Map is a key-value store that can contain any value as a key.
map = %{:hello => "World!", 32 => :age, "height" => 190.47}

# If the key is an Atom, you can access its value using the dot-notation
map.hello # => "World!"

# Otherwise, you need to use the bracket-notation
map[32] # => :age
map["height"] # => 190.47

# Or use a helper function
Map.get(map, 32) # => :age

# A `nil` is returned if the key does not exist
map["foo"] # => nil

# Delete one key from the map
Map.delete(map, :hello)
# => %{32 => :age, "height" => 190.47}

# Or delete multiple keys at once
Map.drop(map, [:hello, 32])
# => %{"height" => 190.47}

  
    
  
  Maps with Atom keys


If your map contains only Atom keys, you can replace the arrow => with a semicolon :
map = %{hello: "World", age: 32, height: 190.47}
map.age
32
The advantage of this notation is that Elixir raises an error if you try to access a key that doesn't exist:
map = %{hello: "World", age: 32, height: 190.47}
map.agee
** (KeyError) key :agee not found in: %{hello: "World", age: 32, height: 190.47}. Did you mean:

      * :age


  
    
  
  Updating a Map


There are a few ways to update a value in a map. They all have their pros and cons:
map = %{age: 32, name: "Peter"}

# This notation updates a map value:
Map.put(map, :age, 33)
# => %{name: "Peter", age: 33}

# But it also adds non-existent keys, which is bad if we make a typo:
Map.put(map, :aeg, 33)
# => %{name: "Peter", age: 32, aeg: 33}

# This notation does not add non-existent keys,
# but it stays silent if we make a typo.
Map.replace(map, :aeg, 33)
# => %{name: "Peter", age: 32}

# This version raises an error instead:
Map.replace!(map, :aeg, 33)
** (KeyError) key :aeg not found in: %{name: "Peter", age: 32}
    (stdlib 6.0) :maps.update(:aeg, 33, %{name: "Peter", age: 32})
Lastly, there is a shorthand notation which comes in handy if you don't want to "pipe" (we'll discuss it later) your output.
map = %{age: 32, name: "Peter"}

# The shorthand notation only updates existing keys:
%{map | age: 33, name: "Pietah"}
# => %{age: 33, name: "Pietah"}

# And raises if the key does not exist
%{map | height: 190.47}
** (KeyError) key :height not found in: %{name: "Peter", age: 32}
    (stdlib 6.0) :maps.update(:height, 190.47, %{name: "Peter", age: 32})
    (stdlib 6.0) erl_eval.erl:465: anonymous fn/2 in :erl_eval.expr/6
    (stdlib 6.0) lists.erl:2146: :lists.foldl/3
    (stdlib 6.0) erl_eval.erl:462: :erl_eval.expr/6
    (elixir 1.17.1) src/elixir.erl:364: :elixir.eval_forms/4
    (elixir 1.17.1) lib/module/parallel_checker.ex:112: Module.ParallelChecker.verify/1
    lib/livebook/runtime/evaluator.ex:629: anonymous fn/3 in Livebook.Runtime.Evaluator.eval/4
    (elixir 1.17.1) lib/code.ex:621: Code.with_diagnostics/2

  
    
  
  Caveats


Map keys are not ordered and if you iterate over the elements of a map, you cannot expect that the "first" key will always be the same.
map = %{"hello" => "World", "age" => 32, "height" => 190.47}
Enum.map(map, fn {key, _value} -> IO.inspect(key) end)
"age"
"height"
"hello"
Maps cannot have duplicate keys. If you define duplicate keys, the latter key will overwrite the earlier key and a warning will be shown:
%{name: "Peter", age: 32, name: "Pietah"}
%{name: "Pietah", age: 32}
warning: key :name will be overridden in map


  

    
Keyword Lists
    

A Keyword list is a list of key-value pairs that has three special characteristics:
	Keys must be Atoms.
	The key-value pairs are ordered.
	Keys can be duplicated.

Here's how to use keyword lists:
keyword = [name: "Peter", age: 32, name: "Pietah"]

# Access an element using the bracket notation
keyword[:age] # => 32

# But it will only return the first key
keyword[:name] # => "Peter"

# You can also use a helper function
Keyword.get(keyword, :age) # => 32

# Replace an existing value with a new one
Keyword.replace(keyword, :age, 33)
# => [name: "Peter", age: 33, name: "Pietah"]

# But when you update a duplicate key, it will only keep the new value and remove the rest
Keyword.replace(keyword, :name, "Pieter")
# => [name: "Pieter", age: 32]

# Delete all pairs that have a given key
Keyword.delete(keyword, :name)
# => [age: 32]

# Only delete the first occurrence
Keyword.delete_first(keyword, :name)
# => [age: 32, name: "Pietah"]


  

    
Modules
    

A Module is a wrapper for functions. If you come from Object-oriented Programming (OOP), they might look a lot like classes, but they are not. They merely help you to organize your code.
Here's how you define a module:
defmodule RunElixir.Profile do

  # This is a 'Module Attribute'. It is a constant value that is set at compile time.
  # You cannot access it from outside the module.
  @legal_age 18

  # But if you need to access the module attribute from outside the module,
  # it is common to write a small public function that returns it:
  def legal_age, do: @legal_age

  # This is a public function. You can call it from outside the module.
  def adult?(age) do
    if age_valid?(age) do
      age >= @legal_age
    else
      raise "Invalid age #{age}"
    end
  end

  # This is a private function. You can call it only from inside the module.
  defp age_valid?(age) do
    age >= 0
  end
end
You can call a module function like this:
RunElixir.Profile.adult?(18) # => true
RunElixir.Profile.adult?(17) # => false
RunElixir.Profile.adult?(-1)
** (RuntimeError) Invalid age -1
    #cell:ephnppjvminxbiag:13: RunElixir.Profile.adult?/1

  
    
  
  alias/2, import/2, and defdelegate/2


You can call the functions of a module from another module in a few ways:
defmodule RunElixir.Checker do

  # Either you call the module using its full namespace:
  def check_age(profile) do
    RunElixir.Profile.adult?(profile.age)
  end

  # Or, you alias its name:
  alias RunElixir.Profile # <- This and import usually go to the top of the module.

  def check_age_aliased(profile) do
    Profile.adult?(profile.age)
  end

  # Or, you import specific or all functions:
  import RunElixir.Profile, only: [adult?: 1] # Remove the 'only: ...' to import all functions

  def check_age_imported(profile) do
    adult?(profile.age)
  end

  # You can also delegate a function call to another module
  # You can't change the arguments, but you can change the function name.
  defdelegate check_adult?(age), to: RunElixir.Profile, as: :adult?
end

RunElixir.Checker.check_age(%{age: 18}) # => true
RunElixir.Checker.check_age_aliased(%{age: 18}) # => true
RunElixir.Checker.check_age_imported(%{age: 18}) # => true
RunElixir.Checker.check_adult?(18) # => true


  

    
Structs
    

A Struct is an extension of a Map and is defined inside a Module. It allows you to define fields at compile-time, require certain fields, set default values, and it raises if you try to set a field that doesn't exist. You can define a struct with defstruct/1.
defmodule RunElixir.Profile do
  # Make the :name field required
  @enforce_keys [:name]
  defstruct [:name, :age, status: :active] # <- Set the default value of :status to :active

  # You can match against the struct like this.
  # This function raises if it receives a plain map and not a struct.
  def adult?(%RunElixir.Profile{age: age}) do
    age >= 18
  end
end
You can create a struct like this:
struct = %RunElixir.Profile{name: "Peter", age: 21}

# You can access the struct's fields using the dot-notation
struct.name # => "Peter"
struct.status # => :active
Structs give you some nice checks for free, like an exception if you try to set a field that doesn't exist.
struct = %RunElixir.Profile{height: 190}
** (KeyError) key :height not found
    expanding struct: RunElixir.Profile.__struct__/1
Structs can also require fields and raise if they are not set.
struct = %RunElixir.Profile{age: 21}
** (ArgumentError) the following keys must also be given when building struct RunElixir.Profile: [:name]
    expanding struct: RunElixir.Profile.__struct__/1


  

    
Functions
    

Since Elixir is a functional programming language, functions are first-class citizens and the foundation for everything you build. You already saw how to define a function inside a module, but let's elaborate on this a bit.
You can define multiple function "clauses" that match against certain values. Function clauses are evaluated top-to-bottom, so the function clause at the top is evaluated first. If it doesn't match, the second function clause is evaluated, and so on. If one function matches, all functions that come thereafter are ignored.
defmodule RunElixir.Checker do

  # This clause will be evaluated first.
  # It matches against integer values from 18 to 150.
  # The >= and <= operators are called 'Guards'.
  def adult?(age) when is_integer(age) and age >= 18 and age <= 150 do
    true
  end

  # This clause will match integer values from 0 to 17.
  # If 'age' was '18', the function clause above would have matched
  # and this function clause would never get evaluated.
  def adult?(age) when is_integer(age) and age >= 0 do
    false
  end

  # This function will match all values that didn't match the function clauses above.
  # This could be integer values outside the range from 0 to 150 or other data types.
  #
  # This is also how you write a one-line function without the do ... end notation.
  def adult?(age), do: raise "Invalid age: #{age}"
end

  
    
  
  Default values


You can define default values using the \\ symbol, like this:
defmodule RunElixir.Profile do
  @legal_age 18

  def adult?(age, legal_age \\ @legal_age) do
    age >= legal_age
  end
end
Now, you can either use the default value or overwrite it like this:
RunElixir.Profile.adult?(18) # => true
RunElixir.Profile.adult?(18, 21) # => false
If you want to define a default value for multiple function clauses, you need to create a function head, like this:
defmodule RunElixir.ProfileChecker do
  @legal_age 18

  # This is the function head which defines the default value for all clauses. It has no body.
  def adult?(age, legal_age \\ @legal_age)

  def adult?(age, legal_age) when age >= legal_age do
    true
  end

  def adult?(age, legal_age) when age >= 0 and age < legal_age do
    false
  end

  def adult?(age, _legal_age), do: raise("Invalid age: #{age}")
end

# Again, you can either use the default value or overwrite it when you call the function:
RunElixir.ProfileChecker.adult?(18) # => true
RunElixir.ProfileChecker.adult?(18, 21) # => false

  
    
  
  Function Arity


Functions have an arity, which describes how many arguments they expect. For example, our adult? function has an arity of 2 because it expects two input arguments, so you would identify the function as adult?/2.
However, one of the arguments has a default value, so we actually define two functions, one with an arity of 1 (plus the default value but it doesn't count) and one with an arity of 2 if we overwrite the default value.
If you list all functions of the Profile module, you'd see that it has two adult? functions:
RunElixir.Profile.__info__(:functions)
[adult?: 1, adult?: 2]

  
    
  
  Anonymous Functions


You can define an anonymous function with fn arguments -> body end.
add = fn a, b -> a + b end
add.(1, 2) # => 3

# A function without arguments
ran = fn -> Enum.random(1..100) end
ran.() # => 20

# You can also use anonymous functions as 'Closures' because they can 'close'
# around variables defined in the same scope and use them later.
value = 20
lazy_evaluate = fn div -> value * 10.0e10 / div end
lazy_evaluate.(2048) # => 976562500.0

# Even if you change the local variable after you define the anonymous function,
# the function will keep the old value.
value = 30
lazy_evaluate.(2048) # => 976562500.0

# A shorthand notation for anonymous functions:
# &1 is the first argument, &2 the second, and so on.
fun = &(&1 + &2)
fun.(4, 5) # => 9
You can assign an anonymous function to a variable and pass it around as an argument too. This comes in handy if you work with callbacks. In this example, we fetch a dad joke from an API using the HTTP library Req and execute the callback function with the response.
# First, we install the Req library.
Mix.install([{:req, "~> 0.5.0"}])

defmodule RunElixir.Jokes do
  def get_dad_joke(callback_fn) do
    # This will execute an anonymous function in an async process and
    # execute the callback function with the result.
    # We will discuss async functions later on.
    spawn(fn ->
      joke = Req.get!("https://icanhazdadjoke.com", headers: [accept: "text/plain"])
      callback_fn.(joke)
    end)
  end
end
Now, let's see how you can pass an anonymous function as an argument.
callback_fn = fn
  # You can define multiple function clauses also in anonymous functions:
  %Req.Response{status: 200, body: joke} -> IO.inspect("Here's a dad joke for you: #{joke}")
  %Req.Response{status: status, body: message} -> IO.inspect("Oh no! An error occurred #{status} - #{message}")
end

RunElixir.Jokes.get_dad_joke(callback_fn)
"Here's a dad joke for you: Why does Han Solo like gum? It's chewy!"

  
    
  
  Return values


There is no explicit return statement in Elixir and you cannot return early from a function. A function always returns the last statement in its body.
fun = fn age ->
  if age >= 18, do: :adult

  :minor
end

fun.(18) # => :minor
fun.(17) # => :minor
1: :minor
2: :minor
The default return value for a function is nil, so if you don't specify a return value explicitly, Elixir will return nil for you. This might surprise you in some cases:
empty_fun = fn -> end

fun = fn age ->
  # The 'if ... end' statement will return 'nil' if it doesn't evaluate to 'true'
  # and you don't provide an 'else' clause.
  if age >= 18 do
    :adult
  end
end

empty_fun.() # => nil

fun.(18) # => :adult
fun.(17) # => nil
The only exception to this is if a function raises an exception. In that case, the rest of the function is not executed:
raises = fn ->
  raise "Boom"

  :return_something
end

raises.() |> IO.inspect()
** (RuntimeError) Boom


  

    
Pipes
    

Elixir makes data transformation extremely easy because you can pass the result of one function into the next by using the pipe |> operator:
add = fn a, b -> a + b end
mult = fn a, b -> a * b end

2
|> add.(3)
|> mult.(5)
|> IO.inspect(label: "Result")
Result: 25

  
    
  
  then/2


A pipe always passes the result of a function into the next one as the first argument. You can use then/2 if you want to change the argument position or modify the result before passing it into the next function.
add = fn a, b -> a + b end
mult = fn a, b -> a * b end

2
|> add.(3)
|> then(fn result ->
  result = result * 2
  mult.(1024, result)
end)
|> IO.inspect(label: "Result")
Result: 10240

  
    
  
  tap/2


The tap/2 helper allows you to 'tap into' a pipe without modifying its current value. It's useful for executing synchronous side effects without changing the value.
add = fn a, b -> a + b end
mult = fn a, b -> a * b end

2
|> add.(3)
|> tap(fn result ->
  IO.inspect(result, label: "Current")
  result * 2 # <- The result of this equation is ignored.
end)
|> mult.(5)
|> IO.inspect(label: "Result")
Current: 5
Result: 25


  

    
Pattern Matching
    

Pattern matching is extremely powerful in Elixir and you'll use it all the time. Let's see some examples:

  
    
  
  Match against Basic Types


# You can match against many data structures, like a map.
# This is a partial match, meaning the 'height' value gets ignored.
%{age: age, name: name} = %{age: 32, name: "Peter", height: 190.47}
IO.inspect([age, name], label: 1)

# Or a Tuple. You can't match partially though.
{a, _, c} = {1, 2, 3}
IO.inspect([a, c], label: 2)

# Or a Keyword list, but you can't match partially either.
[a: a, b: _b, c: c] = [a: 1, b: 2, c: 3]
IO.inspect([a, c], label: 3)

# Or a list, but again, not partially.
[a, b, _] = [1, 2, 3]
IO.inspect([a, b], label: 4)

# But you can match against the head (first element) and tail (the rest) of a list.
[head | tail] = [1, 2, 3]
IO.inspect(head, label: "5 - head")
IO.inspect(tail, label: "5 - tail")
1: [32, "Peter"]
2: [1, 3]
3: [1, 3]
4: [1, 2]
5 - head: 1
5 - tail: [2, 3]

  
    
  
  Caveat: Matching against empty Maps


Watch out if you want to match against an empty map, because Elixir will match any map with an empty map, even if the map isn't empty.
defmodule RunElixir.Match do
  def empty?(%{}), do: true
  def empty?(_), do: false
end

RunElixir.Match.empty?(%{}) # => true
RunElixir.Match.empty?(%{a: 1, b: 2}) # => true

# You can either use the 'map_size/1' guard

defmodule RunElixir.MatchSize do
  def empty?(map) when map_size(map) == 0, do: true
  def empty?(_), do: false
end

RunElixir.MatchSize.empty?(%{}) # => true
RunElixir.MatchSize.empty?(%{a: 1, b: 2}) # => false

# Or a comparison to match against an empty map

defmodule RunElixir.MatchEqual do
  def empty?(map) when map == %{}, do: true
  def empty?(_), do: false
end

RunElixir.MatchEqual.empty?(%{}) # => true
RunElixir.MatchEqual.empty?(%{a: 1, b: 2}) # => false

  
    
  
  Match in Function Clauses


# You can match (partially) in (anonymous) function clauses.
#
# This here defines an anonymous function with two clauses.
# It's the same as defining two function clauses in a module
# like we've done with 'def empty?()' above, just that
# the function isn't wrapped in a module.
fun = fn
  %{status: status} -> status
  %{age: age} -> age
end

fun.(%{status: :active, height: 190.47}) # => :active
fun.(%{age: 32, height: 190.47}) # => 32

# Or in Module functions
defmodule RunElixir.Profile do
  def details(%{status: status}), do: status
  def details(%{age: age}), do: age
end

RunElixir.Profile.details(%{status: :inactive, height: 190.47}) # => :inactive
RunElixir.Profile.details(%{age: 30, height: 190.47}) # => 30

  
    
  
  Match against Strings


# You can match against the end of a string
"My name is " <> name = "My name is Batman"

IO.inspect(name, label: 1)

# Or you can match against specific bytes
#
# This is extremely useful for deconstructing a string into fixed-size byte chunks.
# For example, to implement a communication protocol for which you know that
# the first 3 bytes are one argument, and the next 8 bytes are the second, and so on.
<<article::binary-size(3), _space::binary-size(1), name::binary-size(6), rest::binary>> = "The Batman is my name"

IO.inspect(article, label: "2 - article")
IO.inspect(name, label: "2 - name")
IO.inspect(rest, label: "2 - rest")
1: "Batman"
2 - article: "The"
2 - name: "Batman"
2 - rest: " is my name"

  
    
  
  The Pin Operator


The Pin ^ operator allows you to match against the value of an existing variable:
name = "Peter"

fun = fn
  %{name: ^name} -> "That's me!"
  %{name: _name} -> "That's not me."
end

fun.(%{name: "Peter"}) # => "That's me!"
fun.(%{name: "Bob"}) # => "That's not me."

  
    
  
  Assignment is Matching


Fun fact: In Elixir there is no such thing as a "variable assignment". It's match! When you assign a value to a variable, you actually "match" the value against the variable and because that match is always true, the variable will continue to hold the value. This becomes clearer when we pin the variable:
a = 1
a = 2
^a = 3
** (MatchError) no match of right hand side value: 3
    (stdlib 6.0) erl_eval.erl:652: :erl_eval.expr/6
Interesting! When we pin the variable a and try to assign a new value to it, it raises a MatchError! This happens because a holds the value 2 when we try to match it agains the value 3, which creates a mismatch (Got it? Mis-match! 😬). This shows that the = operator is actually a match, not an assignment. Today you learned!


  

    
Control Flow
    

Elixir offers four control flow structures: if/unless, case, cond, and with.

  
    
  
  if/2 and unless/2


If-else works as you'd expect. If your condition evaluates to truthy (not nil or false), the code block is executed and the result is returned. If the condition evaluates to falsy (nil or false), the else block is executed if it exists; otherwise, nil is returned.
The reverse form of if/2 is unless/2. If your unless condition evaluates to truthy, the else block is executed. If the condition is falsy, the first block is executed. It is common to choose if over unless because it doesn't make your brain hurt, unless you want to execute a side effect or throw an exception.
age = 17

result =
  if age >= 18 do
    :adult
  else
    :minor
  end

IO.inspect(result, label: 1)

unless age >= 18 do
  raise "A minor tries to access the system!"
end
1: :minor
** (RuntimeError) A minor tries to access the system!

  
    
  
  case/2


The case statement is similar to switch in other languages. It allows you to state multiple clauses and tries to match your conditional against each clause.
active_status = :active

fun =
  fn value ->
    case value do
      %{status: ^active_status} -> :active
      %{status: _other_status} -> :inactive

      %{age: nil} -> :age_missing
      %{age: age} when is_integer(age) and age >= 18 -> :adult
      %{age: age} when is_integer(age) -> :minor

      # An optional fallback which always matches.
      # Without it, the case statement would raise an exception if no clause matches.
      _ -> :default
    end
  end

fun.(%{status: :active}) # => :active
fun.(%{status: :foobar}) # => :inactive

fun.(%{age: nil}) # => :age_missing
fun.(%{age: 18}) # => :adult
fun.(%{age: 17}) # => :minor

fun.(nil) # => :default

  
    
  
  cond/1


The cond construct allows you to evaluate multiple clauses and return from the first that evaluates to truthy. It is especially useful if you need to execute a function in a clause.
adult? = fn age -> age >= 18 end

fun =
  fn value ->
    cond do
      value == :active -> :active
      is_integer(value) && adult?.(value) -> :adult
      is_integer(value) -> :minor
      # The optional fallback clause must always evaluate to a truthy value.
      # If you don't give this and no other clause matches, Elixir raises an exception.
      true -> :default
    end
  end

fun.(:active) # => :active
fun.(18) # => :adult
fun.(17) # => :minor
fun.(nil) # => :default

  
    
  
  with/1


The with/1 statement is useful to return early from a sequence of steps if one step fails. If you find yourself writing nested if-else or case statements, you probably want to use with/1 instead.
adult? = fn age -> age >= 18 end
details = %{name: "Peter", age: 32}

# Instead of writing nested if- or case-statements like this:
check_access = fn details ->
  case details do
    %{age: age} ->
      if adult?.(age) do
        :ok
      else
        {:error, :not_adult}
      end

    _ ->
      {:error, :age_missing}
  end
end

# Rather use one with-statement like this:
check_access_refactored = fn details ->
  with %{age: age} <- details,
       true <- adult?.(age) do
    :ok
  else
    # Gets executed if any of the matches above fails.
    %{} -> {:error, :age_missing}
    false -> {:error, :not_adult}
  end
end
To match inside the else-block isn't great though. What if two matches can return false? How would you know which match failed?
It is good practice to move steps step into small helper functions. This allows you to return a specific error message depending on which step failed and it makes the with-statement more readable. It also makes it easy to see the "happy path" of your function and all its error cases.
defmodule RunElixir.Permissions do
  def check_access(details) do
    with {:ok, age} <- get_age(details),
         :ok <- check_adult(age) do
      :ok
    end
  end

  # Moving each step into a small helper function gives you the flexibility
  # to decide which error to return inside the function.
  defp get_age(details) do
    case details do
      %{age: age} when is_integer(age) -> {:ok, age}
      %{age: _age} -> {:error, :age_invalid}
      _ -> {:error, :age_missing}
    end
  end

  defp check_adult(age) do
    if age >= 18, do: :ok, else: {:error, :not_adult}
  end
end

RunElixir.Permissions.check_access(%{age: 32}) # => :ok
RunElixir.Permissions.check_access(%{age: nil}) # => {:error, :age_invalid}
RunElixir.Permissions.check_access(%{name: "Peter"}) # => {:error, :age_missing}
RunElixir.Permissions.check_access(%{age: 17}) # => {:error, :not_adult}


  

    
Loops
    

You can iterate over lists, tuples, keyword lists, and maps in a few different ways. Here are the most common ones:

  
    
  
  The for/1 Comprehension


# You can iterate over a range ascending
result = for idx <- 1..10//1, do: idx
IO.inspect(result, label: "range - asc")

# Or descending. Note the step of -2.
# If you don't set the step, the loop will stop after the first element.
result = for idx <- 10..1//-2, do: idx
IO.inspect(result, label: "range - desc")

# You can iterate through a list of elements
result = for el <- ["Peter", 32, 190.47, :active], do: el
IO.inspect(result, label: "elements")

# Or you can have a nested for-loop
result = for x <- [1, 2], y <- [4, 5], do: [x: x, y: y]
IO.inspect(result, label: "nested")
range - asc: [1, 2, 3, 4, 5, 6, 7, 8, 9, 10]
range - desc: [10, 8, 6, 4, 2]
elements: ["Peter", 32, 190.47, :active]
nested: [[x: 1, y: 4], [x: 1, y: 5], [x: 2, y: 4], [x: 2, y: 5]]

  
    
  
  Enum helpers


You will mostly use the Enum helper functions to iterate and modify collections. Here are the most used ones:
# Apply a function to every element and return the result.
Enum.map([1, 2, 3], fn x -> x * 2 end)
# => [2, 4, 6]

# Apply a function, but ignore the result.
Enum.each([1, 2, 3], fn x -> x * 2 end)
# => :ok

# Apply a function and add the result to an accumulator.
Enum.reduce([1, 2, 3], 0, fn x, acc -> acc + x end)
# => 6

# Only keep the elements for which the function evaluates to truthy and filter out the rest.
Enum.filter([1, 2, 3], fn x -> x == 2 end)
# => [2]

# Reject/remove an element if the function evaluates to truthy.
Enum.reject([1, 2, 3], fn x -> x == 2 end)
# => [1, 3]

  
    
  
  Recursive Functions


A common pattern in functional programming is to have recursive loops using function clauses. Elixir uses tail call optimisation to keep the memory overhead low and to avoid stack overflows.
defmodule RunElixir.Loop do
  def loop(collection, fun), do: do_loop(collection, [], fun)

  # Applies an anonymous function to each element and returns the result as list.
  defp do_loop([], result, _fun), do: Enum.reverse(result)
  defp do_loop([value | rest], result, fun), do: do_loop(rest, [fun.(value) | result], fun)
end

RunElixir.Loop.loop([1, 2, 3], &(&1 * 2))
# => [2, 4, 6]
Our loop/2 function executes the do_loop/3 private function which calls itself recursively until it has traveresed the entire list. With every call, the second clause of do_loop/3 removes one element from the list, applies the anonymous function to it, and adds the result to the results list. When the list of elements is exhausted, the first do_loop/3 clause matches. It reverses the results and returns them.


  

    
Create a Phoenix Project
    

A fast way to explore Elixir further is to create a website using the Phoenix framework.

  
    
  
  Prerequisites


	You have installed Elixir (If not, see here for instructions)
	You have a Postgres database running (If not, see here for instructions)


  
    
  
  Create a Website


You can create and run a new website with only a few commands:
# Install the Phoenix project generator
mix archive.install hex phx_new

# Create a new project
mix phx.new demo --install

# Navigate into the new project
cd demo

# Create the database
mix ecto.create

# And start the server
mix phx.server

Now, go to localhost:4000 and you'll see a welcome page!

  
    
  
  Build a ToDo App in 1 Minute


Let's add some reactive content to the project! We can build a ToDo app in under a minute. Just run the following in your terminal:
# Generate a ToDo database schema and add a UI to manage them
mix phx.gen.live ToDos ToDo todos name:string done:boolean deadline:utc_datetime

Then, copy and paste the live routes into your lib/demo_web/router.ex
scope "/", DemoWeb do
  pipe_through :browser

  get "/", PageController, :home

  # Add these here👇
  live "/todos", ToDoLive.Index, :index
  live "/todos/new", ToDoLive.Index, :new
  live "/todos/:id/edit", ToDoLive.Index, :edit

  live "/todos/:id", ToDoLive.Show, :show
  live "/todos/:id/show/edit", ToDoLive.Show, :edit
end
Now, go to localhost:4000/todos. It will ask you to run the migrations. Just click on Run migrations for repo.
You should now see an empty list of ToDos. Just click on New To do to add your first todo! Try editing and deleting some todos.
You might wonder why you don't see page reloads when you add or remove todos. That's because we use Phoenix LiveView, which allows us to write reactive but server-rendered frontends in pure Elixir!
Check out the Phoenix LiveView docs to learn more or watch the creator of Phoenix, Chris McCord, build a real-time Twitter clone in just 15:


  

    
Install Dependencies
    

You can install dependencies from a few sources, like the Elixir package manager Hex.pm, GitHub, a Git repository, or a local path. You only need to add them to the deps/0 function in your mix.exs file.
Let's add the HTTP library, Req:
# In your mix.exs file.

defp deps do
  [
    # Other dependencies
    # ...
    # Add this to download a dependency from hex.pm
    {:req, "~> 0.5.6"},
    # Or fetch a dependency directly from GitHub
    {:oban, github: "sorentwo/oban"},
    # Or from a Git repo
    {:prom_ex, git: "https://github.com/akoutmos/prom_ex.git"}
    # Or from a local path (commented out because it doesn't exist)
    # {:your_lib, path: "../relative/path/to/your/lib"}
  ]
end
Now, run mix deps.get to fetch the dependencies.
Let's start our website in an interactive shell to test the library. Run this in your terminal:
iex -S mix phx.server
Now you can execute Elixir code inside your project. Run this command to fetch a dad joke from the internet:
iex> Req.get!("https://icanhazdadjoke.com", headers: [accept: "text/plain"]).body
"I think circles are pointless."
And that's it! You've successfully added and tested a library :)


  

    
Deploy your Project
    

Once you're ready to deploy your Phoenix project, try out one of the following deploy guides:
	Fly.io
	Gigalixir
	Heroku



  

    
Continue Learning
    

Aside from the resources below, the best place to continue learning is the official Elixir documentation.
The following list is non-exhaustive and has no particular order.

  
    
  
  Books


	Elixir in Action
	PragProg Elixir Books
	Elixir Patterns
	Joy of Elixir


  
    
  
  Video Courses


	Indie Courses Catalogue
	ElixirCasts
	Alchemist Camp
	TechSchool
	LearnElixir
	Phoenix LiveView
	Grox.io
	Pragmatic Studio's Elixir, Phoenix, and LiveView courses


  
    
  
  Interactive Guides


	Exercism
	Elixir School


  
    
  
  Others


	Elixir Koans on GitHub
	The official learning recommendations



  

    
Join the Community
    

The Elixir community hangs out in a few different places. I ranked them by guesstimated size:
	Elixirforum
	Elixir Slack
	Elixir Discord
	Reddit r/elixir
	Twitter (search for #MyElixirStatus or follow the official account @elixirlang)
	LinkedIn Group
	Mastodon: genserver.local



  

    
Stay Informed
    

Stay up-to-date with this non-exhaustive list of Elixir content channels in no particular order:

  
    
  
  Newsletters


	Elixir Radar
	Elixir Weekly
	Elixir Merge


  
    
  
  Podcasts


	Thinking Elixir
	Elixir Wizards
	Elixir Mix
	Beam Radio
	Elixir Em Foco (Portuguese)
	Elixir Newbie


  
    
  
  Blogs


	Peter Ullrich (me!) and Indie Courses (my startup!)
	Official Elixir Blog
	Dashbit
	Fly.io
	Elixirforum Blog Thread


  
    
  
  YouTube Channels


	Elixir Mentor
	Daniel Bergholz
	German Velasco
	Andrew Stewart
	ElixirConf (Conference Talks)
	CodeSync (Conference Talks)



  

    
Find a Job
    

You can find the most Elixir job postings here:
	Elixir Jobs
	Elixir Radar Job Board
	Elixirforum: Elixir Jobs Topic
	LinkedIn
	Elixir Slack #jobs or #looking_for_contract
	Twitter - Search for or tweet with the hashtag #MyElixirStatus



  

    
Find Libraries
    

You will run into the most popular libraries pretty quickly since the Elixir community tends to not create 10-times the same thing, but rather gather around one or two libraries that get the job done.
However, if you want to browse a bit, these are good resources:
	Essential Libraries for Elixir Developers
	Awesome Elixir
	Elixir Toolbox



  

    
Logger
    


  
    
  
  Section


Elixir provides a single interface for creating logs in your application through the Logger module.
defmodule RunElixir.LogExample do

  # You always have to `require` the Logger first.
  # Elixir will warn you if you forget this.
  require Logger

  def test() do
    # Logger supports all 7 system log levels:
    Logger.debug("This is a debug message")
    Logger.info("This is an info message")
    Logger.notice("This is a notice message")
    Logger.warning("This is a warning message")
    Logger.error("This is an error message")
    Logger.critical("This is a critical message")
    Logger.alert("This is an alert message")
    Logger.emergency("This is an emergency message")
  end
end

RunElixir.LogExample.test()
10:31:41.197 [debug] This is a debug message
10:31:41.198 [info] This is an info message
10:31:41.198 [notice] This is a notice message
10:31:41.198 [warning] This is a warning message
10:31:41.198 [error] This is an error message
10:31:41.198 [critical] This is a critical message
10:31:41.198 [alert] This is an alert message
10:31:41.198 [emergency] This is an emergency message
The debug, info, warning, and error levels are widely used in Elixir applications and you'll encounter them all the time. critical messages are usually logged only by the system or the virtual machine and I've yet to see a notice, alert, or emergency log.

  
    
  
  Caveat


You have to watch out when you want to log variables. Not all variables implement the String.Chars protocol, which means Elixir doesn't know how to convert them to a string. If you try to log them anyway, you'll encounter an error.
require Logger

Logger.info("My map: #{%{a: 1}}")
** (Protocol.UndefinedError) protocol String.Chars not implemented for %{a: 1} of type Map
    (elixir 1.17.2) lib/string/chars.ex:3: String.Chars.impl_for!/1
    (elixir 1.17.2) lib/string/chars.ex:22: String.Chars.to_string/1
You can prevent this issue by wrapping your variables with inspect/2 first. The Inspect protocol can "stringify" almost all variable types, so it's a pretty safe option to use.
require Logger

Logger.info("My map: #{inspect(%{a: 1})}")
10:40:10.782 [info] My map: %{a: 1}

  
    
  
  Configuration


The Logger offers extensive configuration options, but you can configure them only for an Elixir application, not a script or Livebook as shown above. If you followed the Create a Phoenix Project guide, you can configure the logger in the config/(config|dev|prod|test).exs files.
You can set global options that should apply to all environments in the config/config.exs file and add different options in the test.exs, dev.exs, or prod.exs files. Keep in mind that any configuration you put in the test|dev|prod.exs files will overwrite existing options from the config.exs configuration.
This is an example of how you can overwrite the global configuration for the dev environment.
# e.g. in config/config.exs

config :logger, :console,
  format: "\n$date $time [$level] $message\n",
  level: :info

# e.g. in config/dev.exs

# This configuration will overwrite the `level`-option
# from the `config.exs` file, but keep its `format`-option.
config :logger, :console, level: :debug
Now, Elixir will log all messages, including debug messages, in your local development environment using the format defined in config.exs. You can find all format options here. If you run your application with mix phx.server, you'll see the log messages have now the following format.
2024-09-16 10:57:05.096 [info] log message here
2024-09-16 10:57:06.030 [info] another log message


  

    
Testing
    

Elixir comes with a powerful testing framework called ExUnit. It makes writing unit tests incredibly simple. For end-to-end tests, the go-to library is Wallaby, but if you use LiveView, you should be able to write most of your integration tests in ExUnit without having to start a browser. Let's have a look at how to use ExUnit and Wallaby for our tests.

  
    
  
  Unit Tests


Unit tests in ExUnit usually have the following structure and are stored in the test folder. They need to have the *_test.exs file ending, otherwise the test runner will ignore them.
# To run the tests, you need to start ExUnit first.
# In your application, this is usually executed in the "test/test_helper.exs" file.
ExUnit.start()

# lib/example.ex
defmodule RunElixir.Example do
  @moduledoc "The module we want to test."

  def div(nominator, denominator), do: nominator / denominator
end

# test/example_test.exs
defmodule RunElixir.ExampleTest do
  # Defines this module as a test case and instructs the test runner
  # to run the tests in parallel with other tests.
  use ExUnit.Case, async: true

  alias RunElixir.Example

  # A setup callback that runs before every test.
  #
  # You can also use `setup_all` to run a test setup
  # only once for all tests.
  setup do
    # The "context" map returned by the setup and setup_all callbacks
    # is passed to every describe and test block as an argument.
    %{number: 2}
  end

  # descibe/1 blocks are useful for wrapping tests for a function or feature
  describe "div/2" do

    # describe blocks can have their own setup and setup_all callbacks
    # which run only for the tests inside the describe-block
    setup %{number: number} do
      # You can modify the context map if you want.
      %{number: number + 2, divisor: 4}
    end

    test "returns a fraction", %{number: number, divisor: divisor} do
      # assert expects a "truthy" value and otherwise fails the test.
      assert Example.div(number, divisor) == 0.5
    end

    test "throws an exception if divided by Zero", %{number: number} do
      # You can assert many different scenarios.
      # Check the `ExUnit.Assertions` block for all options.
      assert_raise ArithmeticError, fn ->
        Example.div(number, 0)
      end
    end
  end
end

# You could run the tests with "mix test" in your terminal.

# Alternatively, you can execute tests from an IEx session like this:
ExUnit.run()
Running ExUnit with seed: 576563, max_cases: 20

.

  1) test div/2 see how a test fail looks like (RunElixir.ExampleTest)
     Library/Application Support/livebook/autosaved/2024_09_27/10_48_dvyj/untitled_notebook.livemd#cell:hpn2l3c23pevgamx:55
     Assertion with == failed
     code:  assert Example.div(2, 4) == 1.0
     left:  0.5
     right: 1.0
     stacktrace:
       Library/Application Support/livebook/autosaved/2024_09_27/10_48_dvyj/untitled_notebook.livemd#cell:hpn2l3c23pevgamx:56: (test)

..
Finished in 0.00 seconds (0.00s async, 0.00s sync)
4 tests, 1 failure

  
    
  
  Integration Tests with LiveView


You can write powerful unit tests with ExUnit and easily extend them to integration tests using the LiveView test helpers.
If you generated the ToDo app in chapter Create a Phoenix Project, Phoenix generated a complete test setup and some unit tests for you. You can find the unit tests in test/demo/to_dos_test.exs and test/demo_web/live/to_do_live_test.exs and the test setup in test/test_helper.exs and test/support/(data|conn)_case.ex.
Let's have a look at the "saves new to_do" unit test in test/demo_web/live/to_do_live_test.exs:
defmodule DemoWeb.ToDoLiveTest do
  # This imports common test functions like our assertions,
  # creates a sandboxed database connection, and builds a connection
  # that we can use to load the website.
  # Since our database connections are sandboxes, we can run tests in parallel.
  # That's what the async: true flag does here. If this was async: false,
  # the tests in this file would run sequentially, not in parallel.
  use DemoWeb.ConnCase, async: true

  # These import our LiveView-specific test helpers and assertions
  # and our "fixtures" with which we can quickly create a
  # ToDo database record.
  import Phoenix.LiveViewTest
  import Demo.ToDosFixtures

  @create_attrs %{name: "some name", done: true, deadline: "2024-09-22T12:12:00Z"}
  @invalid_attrs %{name: nil, done: false, deadline: nil}

  # This defines a setup callback we can execute before each test execution.
  # It inserts a ToDo database record and returns a map that is merged into
  # the context map which our tests receive as an argument.
  defp create_to_do(_) do
    to_do = to_do_fixture()
    %{to_do: to_do}
  end

  # We use 'describe' to wrap a group of tests, in this case
  # we wrap all tests related to the Index LiveView which lists all ToDos.
  describe "Index" do
    # We can define one or multiple setup callbacks that are executed
    # before each test unit using `setup`. We could also use `setup_all` if
    # we wanted to execute the callbacks only once for all tests.
    setup [:create_to_do]

    # Each test needs a name and might receive a context map as a second parameter.
    # The context map can be populated in test setup callbacks like `create_to_do/1`.
    test "saves new to_do", %{conn: conn, to_do: to_do} do
      # First, we navigate to the "List ToDos" route and receive a LiveView connection
      # and the rendered html as return values.
      {:ok, index_live, _html} = live(conn, ~p"/todos")

      # Next, we click on the "New To do" button which should open up a modal.
      assert index_live |> element("a", "New To do") |> render_click() =~
               "New To do"

      # We assert that the click on the button patch-navigated us to the modal-showing route.
      # Since it's a patch navigation, we don't reload the LiveView, but only execute the
      # handle_params/3 callback in the LiveView.
      assert_patch(index_live, ~p"/todos/new")

      # We try to insert invalid parameters and assert that
      # the form shows us a "can't be blank" form error.
      assert index_live
             |> form("#to_do-form", to_do: @invalid_attrs)
             |> render_change() =~ "can&#39;t be blank"

      # We fill out the form with valid attributes and submit the form
      # which should create a new ToDo database record.
      assert index_live
             |> form("#to_do-form", to_do: @create_attrs)
             |> render_submit()

      # We assert that we patch-navigate back to the "List ToDos" route
      # after we submitted the form.
      assert_patch(index_live, ~p"/todos")

      # We re-render the page to assert that we see a Flash message
      # and that the name of the new ToDo shows on the page.
      html = render(index_live)
      assert html =~ "To do created successfully"
      assert html =~ "some name"
    end
  end
end
The best thing about running integration tests with LiveView is that you can easily drop down to the database level and check that a record was persisted correctly. Let's extend the test above to test just that.
test "saves new to_do", %{conn: conn, to_do: to_do} do

  # Prior test code omitted ...

  html = render(index_live)
  assert html =~ "To do created successfully"
  assert html =~ "some name"

  # Add the following lines:

  # Fetch all ToDos in the database.
  todos = Demo.ToDos.list_todos()

  # Assert that we now have two ToDos in the database,
  # one from the setup callback and one from submitting the form.
  assert length(todos) == 2

  # Find the new ToDo which we created through the form.
  todo = Enum.find(todos, & &1.id != to_do.id)

  # Assert that all fields were set properly.
  assert todo.name == "some name"
  assert todo.done == true
  assert todo.deadline == ~U[2024-09-22 12:12:00Z]
end
You can run your tests with:
mix test

# Or run the test in only one file
mix test test/demo_web/live/to_do_live_test.exs

# Or run only the test starting in line 26
mix test test/demo_web/live/to_do_live_test.exs:26

# Calculate the test coverage with
mix test --cover

  
    
  
  Notes about Unit Tests


	All test files must end with *_test.exs, otherwise they are not executed.
	You commonly create test files in the folders: test/app_name/ and test/app_name_web and mirror the folder structure of your app. So, if you want to test the file at lib/demo/to_dos.ex, you'd create a test file at test/demo/to_dos_test.exs.
	You can also colocate your test and your code files, so lib/demo/to_dos.ex would have its test file at lib/demo/to_dos_test.exs. You only need to add test_paths: ["test", "lib"] to your project options in mix.exs and copy&paste the test/test_helper.exs to lib/test_helper.exs. That will instruct ExUnit to execute any *_test.exs file in your lib folder as well.


  
    
  
  End-to-end Tests


For end-to-end testing, Wallaby is a popular choice in the Elixir community. It allows you to write tests that simulate user interactions in a real browser, which can be especially useful for testing complex UI interactions or scenarios that are difficult to simulate with LiveView tests alone.
Explaining how to set up and use Wallaby is a bit too complex for this guide, so please have a look at their official documentation.


  

    
Introduction
    

Working with asynchronous processes is Elixir's strong suit because it builds on top of the ~40-year-old virtual machine of Erlang, the BEAM. The BEAM was originally designed in the telecommunication industry and was optimised for parallelism instead of raw speed. It was meant to handle massive amounts of simultaneous connections with high fault tolerance and availability.
That's why the BEAM, and therefore Elixir, comes with a rich set of asynchronous features. They all revolve around a Process and the Actor Model. BEAM processes are lightweight with a small memory footprint (~7kb per process), fast to start and stop, and efficient to schedule. Every process has its own heap and doesn't share memory with other processes. All communication between processes happens through messages, and each process has its own mailbox. With small exceptions, "sending" a message means that the sending process copies the message into the receiving process's mailbox. The receiving process checks its mailbox in a loop and handles new messages separately.
Let's start with the simplest way to start an async process: spawn/1


  

    
spawn/1
    

The spawn/1 function is a quick fire-and-forget way to start an asynchronous process.
fun = fn ->
  # spawn/1 takes an anonymous function and
  # returns the Process Identifier (PID) of the new process.
  #
  # A crash in the spawned process doesn't impact the spawning process.
  pid =
    spawn(fn ->
      IO.inspect("I'm alive!")
      raise "BOOM!"
    end)

  IO.inspect("Started #{inspect(pid)}")
  :timer.sleep(1)
  IO.inspect("Still alive 😎")
end

fun.()
"Started #PID<0.171.0>"
"I'm alive!"
** (RuntimeError) BOOM!
"Still alive 😎"
You can link two processes if you want both to die if one of them dies. So, if a parent process spawns a child process and we link them, the parent dies if the child process dies and vice versa. To "die" in this context means that one process sends an :EXIT message to the other process that contains an exit reason other than :normal.
fun = fn ->
  # Start a parent process unlinked to the outer process,
  # but linked to its child. When the child dies,
  # the parent dies too, but the outer process is unaffected.
  pid =
    spawn(fn ->
      IO.inspect("Parent is alive! Spawning child...")
      spawn_link(fn -> raise "BOOM!" end)
      :timer.sleep(1) # <- Wait for the child to start.
      IO.inspect("Parent is still here!") # <- Will not execute
    end)

  IO.inspect("Started parent at #{inspect(pid)}")
  :timer.sleep(1)
  IO.inspect("Still alive 😎")
end

fun.()
"Started parent at #PID<0.215.0>"
"Parent is alive! Spawning child..."
** (RuntimeError) BOOM!
"Still alive 😎"


  

    
Task
    

Usually, you wouldn't work with spawn/1 directly but use the Task module instead. It offers a range of helper functions for spawning one-off tasks.
# Use `Task.start/1` instead of `spawn/1` to start an unlinked process
# and `Task.start_link/1` instead of `spawn_link/1` to start a linked one.
fun = fn ->
  {:ok, pid} =
    Task.start(fn ->
      IO.inspect("Parent Task started!")
      Task.start_link(fn -> raise "BOOM!" end)
      :timer.sleep(1)
      IO.inspect("Parent Task is still alive!") # <- Won't execute
    end)

  IO.inspect("Parent started at #{inspect(pid)}")
  :timer.sleep(1)
  IO.inspect("Still alive 😎")
end

fun.()
"Parent started at #PID<0.232.0>"
"Parent Task started!"
** (RuntimeError) BOOM!
"Still alive 😎"

  
    
  
  Task.async/1 and Task.await/1


A big advantage of using Task over spawn/1 is that you can easily await the result of the async process using the common async/await notation.
fun = fn ->
  pid =
    Task.async(fn ->
      IO.inspect("Task starts work", label: NaiveDateTime.utc_now())
      :timer.sleep(500)
      :some_result
    end)

  IO.inspect("Starting other work", label: NaiveDateTime.utc_now())
  :timer.sleep(200) # <- Do some other work
  IO.inspect("Finished other work", label: NaiveDateTime.utc_now())

  # Whenever you're ready, you can await the result.
  result = Task.await(pid, :timer.seconds(1)) # Wait up to 1 second
  IO.inspect("Task returned: #{result}", label: NaiveDateTime.utc_now())
end

fun.()
2024-08-31 15:35:17.500386: "Starting other work"
2024-08-31 15:35:17.500489: "Task starts work"
2024-08-31 15:35:17.700461: "Finished other work"
2024-08-31 15:35:18.000527: "Task returned: some_result"
Keep in mind that calling await/1 in the awaiting process will block the process until the awaited process returns a value or until the timeout is reached (5 seconds by default). If the timeout is reached, both processes will die 💀

  
    
  
  Task.async_stream/3


You can easily apply an operation to a list of elements and collect the result in parallel by using Task.async_stream/3.
elements = ["elixir", "is", "great"]

# Count the characters in each word and return the overall sum.
#
# Task.async_stream/3 runs as many tasks as you have schedulers in parallel.
# This defaults to the number of available CPUs. Each task applies
# the function to only one element and returns the result.
elements
|> Task.async_stream(fn word -> String.length(word) end)
|> Enum.reduce(0, fn {:ok, count}, acc -> count + acc end)
13
Take a second and look at this again.
In just 3 lines of code, we parallelised a task over all available CPUs! 🤯


  

    
GenServer
    

A GenServer is what you commonly use for a process that is either long-running or more complex than a simple one-off task. They are arguably the most common method for doing anything asynchronously in Elixir and you'll encounter them all the time. GenServers can easily send and receive messages - also to themselves - keep long-lasting state, and perform background tasks.
defmodule RunElixir.GameState do
  use GenServer

  require Logger

  # Public functions
  #
  # All these functions are executed in the calling process, not the GenServer process.

  # It's common to write a public helper function to a GenServer process,
  # and required if you start the GenServer from a Supervisor (see next section).
  #
  # You can use `GenServer.start/3` or `GenServer.start_link/3` to start a new process.
  def start_link(args) do
    # The optional 'name'-option registers the GenServer with a unique name
    # that can be used to call the GenServer from anywhere. This is useful
    # for starting Singletons, but will block you from starting more than one
    # process based on this GenServer. That's why it's good practice to
    # make the name configurable through the arguments.
    name = Keyword.get(args, :name, __MODULE__)
    GenServer.start_link(__MODULE__, args, name: name)
  end

  # `GenServer.cast/2` sends a message to the GenServer process
  # and returns `:ok` without waiting for a response.
  def add_points(name_or_pid \\ __MODULE__, points) do
    GenServer.cast(name_or_pid, {:add_points, points})
  end

  # `GenServer.call/2` makes a synchronous call to the GenServer
  # and waits for a response with a default timeout of 5 seconds.
  def end_round(name_or_pid \\ __MODULE__), do: GenServer.call(name_or_pid, :end_round)
  def get_state(name_or_pid \\ __MODULE__), do: GenServer.call(name_or_pid, :get_state)


  # Callbacks
  #
  # All these functions are executed in the GenServer process.

  @initial_state %{
    rounds_played: 0,
    total_points: 0,
    current_points: 0
  }

  # This callback is required. It receives the 'args'-variable from
  # the GenServer.start_link/3 or GenServer.start/3 above and returns
  # the initial state of the process.
  #
  # It can optionally return a third argument with {:continue, tuple_or_atom}
  # which will be handled asynchronously in the `handle_continue/2` callback.
  def init(_args), do: {:ok, @initial_state, {:continue, :log_start}}

  # It is good practice to return quickly from `init/1` and to move
  # heavy workloads into the `handle_continue/2` callback. This way
  # you don't block whatever starts your GenServer (e.g. your application on startup)
  # because - unlike `init/1` - this callback is executed asynchronously
  # to the rest of your system.
  def handle_continue(:log_start, state) do
    Logger.info("GameState has started!")
    {:noreply, state}
  end

  # This callback handles the `cast/2`-call from above.
  # It does not return a value to the calling process.
  #
  # All callbacks will receive and may modify the process state
  # but have to return something as the new process state
  # which will then be passed into the next callback.
  # In this case, we simply return the current state.
  def handle_cast({:add_points, new_points}, state) do
    state = Map.update!(state, :current_points, fn points -> points + new_points end)
    {:noreply, state}
  end

  # These callbacks handle the `call/2`-calls from above.
  # You can define callbacks multiple times and match
  # against the arguments from different `call/2` or `cast/2`-calls.
  #
  # This one updates the process state and returns
  # `{:ok, new_state}` to the calling process.
  def handle_call(:end_round, _from, old_state) do
    %{
      rounds_played: rounds_played,
      total_points: total_points,
      current_points: current_points
    } = old_state

    new_state = %{
      rounds_played: rounds_played + 1,
      total_points: total_points + current_points,
      current_points: 0
    }

    {:reply, {:ok, new_state}, new_state}
  end

  # This one only returns the state.
  # Try to minimise fetching the state by always returning the updated state if you modify it,
  # as we've done in the `handle_call/2` above, because fetching the state repeatedly
  # blocks both the calling and the called process.
  def handle_call(:get_state, _from, state), do: {:reply, state, state}
end
Let's try out our new GameState GenServer!
alias RunElixir.GameState

# Let's start a new GameState process with a custom name.
{:ok, pid} = GameState.start_link([name: :game])

# We can call the GenServer either with its (atom) name or its PID.
GameState.get_state(pid) |> IO.inspect(label: 1)
GameState.add_points(:game, 100) |> IO.inspect(label: 2)
GameState.get_state(:game) |> IO.inspect(label: 3)
GameState.end_round(:game) |> IO.inspect(label: 4)

# We can easily start another process, for example, for another player.
{:ok, pid} = GameState.start_link([name: :new_game])
GameState.get_state(:new_game) |> IO.inspect(label: 5)

# Or, we can start the GenServer under its module name
{:ok, _pid} = GameState.start_link([])
GameState.add_points(50) |> IO.inspect(label: 6)
GameState.end_round() |> IO.inspect(label: 7)
14:49:27.607 [info] GameState has started!
1: %{rounds_played: 0, total_points: 0, current_points: 0}
2: :ok
3: %{rounds_played: 0, total_points: 0, current_points: 100}
4: {:ok, %{rounds_played: 1, total_points: 100, current_points: 0}}

14:49:27.608 [info] GameState has started!
5: %{rounds_played: 0, total_points: 0, current_points: 0}

14:49:27.608 [info] GameState has started!
6: :ok
7: {:ok, %{rounds_played: 1, total_points: 50, current_points: 0}}


  

    
Supervisor
    

A Supervisor allows you to start a predefined list of GenServers and to restart them if they crash. We can also use a DynamicSupervisor if we want to start GenServers dynamically and on demand, or a PartitionSupervisor, if we wanted to group the supervised GenServers by specific keys. But for this example, we'll focus on the "plain" Supervisor.
Let's first define a simple GenServer that we can start and supervise:
defmodule RunElixir.Buffer do
  @moduledoc "Implements a naive Last-in-First-out (LIFO) buffer."

  use GenServer

  require Logger

  def start_link(_args), do: GenServer.start_link(__MODULE__, [])

  def push(pid, message), do: GenServer.cast(pid, {:push, message})
  def pop(pid), do: GenServer.call(pid, :pop)

  # Callbacks

  def init(_args) do
    Logger.info("Buffer started!")
    {:ok, []}
  end

  def handle_cast({:push, message}, state), do: {:noreply, [message | state]}
  def handle_call(:pop, _from, [message | state]), do: {:reply, message, state}
end
Now, let's start two buffers using a Supervisor:
children = [
  %{id: :buffer1, start: {RunElixir.Buffer, :start_link, [nil]}},
  %{id: :buffer2, start: {RunElixir.Buffer, :start_link, [nil]}},
]

{:ok, pid} = Supervisor.start_link(children, strategy: :one_for_one)

Supervisor.count_children(pid) |> IO.inspect(label: 1)

Supervisor.which_children(pid) |> IO.inspect(label: 2)
17:08:53.997 [info] Buffer started! # <- Buffer 1 has started
17:08:53.997 [info] Buffer started! # <- Buffer 2 has started
1: %{active: 2, workers: 2, supervisors: 0, specs: 2}
2: [
  {:buffer2, #PID<0.835.0>, :worker, [RunElixir.Buffer]},
  {:buffer1, #PID<0.834.0>, :worker, [RunElixir.Buffer]}
]
We see that the Supervisor started both buffers successfully. Let's try to add and pop some messages from the first buffer.
[{_name, buffer_pid, _type, _module} | _] = Supervisor.which_children(pid)

RunElixir.Buffer.push(buffer_pid, "Hello")
RunElixir.Buffer.push(buffer_pid, "World")
RunElixir.Buffer.pop(buffer_pid) |> IO.inspect(label: 1)
RunElixir.Buffer.pop(buffer_pid) |> IO.inspect(label: 2)
RunElixir.Buffer.pop(buffer_pid)
1: "World"
2: "Hello"

15:54:56.530 [error] GenServer #PID<0.808.0> terminating
** (FunctionClauseError) no function clause matching in RunElixir.Buffer.handle_call/3

15:54:56.530 [info] Buffer started!
Oh no! We forgot to handle the case when we try to pop a message from an empty buffer. Thats why the buffer crashed when we tried to pop a third message. Luckily, the Supervisor has got our back and restarted the buffer right away, as you can see in the last log line.

  
    
  
  Strategies


We started our children using the :one_for_one strategy, which means that the Supervisor will restart only the crashed GenServer process and won't touch the remaining ones.
We could also use the :one_for_all strategy, which would instruct the Supervisor to terminate all remaining processes and then to restart all processes, including the terminated one.
Lastly, we could use the :rest_for_one strategy, which means that the Supervisor terminates all process that come after the crashed process in the list of children and then restarts the terminated processes, including the crashed one.
You should choose the strategy that fits your use case best, but you'll mostly see the :one_for_one strategy "in the wild".


  

    
Introduction
    

Keeping in-memory state is trivial in Elixir because its virtual machine, the BEAM, was designed to handle short-lived and ephemeral state without a database. We already looked at the GenServer which starts a process that has its own state, but Elixir has even more options to offer. Let's dive in.

An Agent is a specialized GenServer that provides a set of helper functions which make it easy to store and update state in-memory using a process. Agents come in handy if you want a simple GenServer to hold onto state.
defmodule RunElixir.GameState do
  use Agent

  # Public

  @initial_state %{
    rounds_played: 0,
    current_points: 0,
    total_points: 0
  }

  # Starts a new Agent process with an initial state.
  def start_link(_args) do
    Agent.start_link(fn -> @initial_state end)
  end

  # Creates a fire-and-forget message to the Agent.
  #
  # Unlike in a GenServer, we don't have to write callbacks
  # for our events like `handle_cast/2` or `handle_call/3`
  # but can define the callback as an anonymous function.
  #
  # The anonymous function receives the current state
  # and has to return the new state.
  def add_points(pid, points) do
    Agent.cast(pid, fn state ->
      Map.update!(state, :current_points, &(&1 + points))
    end)
  end

  # Returns the current state.
  def get_state(pid), do: Agent.get(pid, fn state -> state end)

  # Ends the round by updating and returning the state.
  #
  # `Agent.get_and_update/2` receives the current state in the
  # anonymous function and has to return a tuple with the state
  # that should be returned to the calling process and
  # the new process state.
  def end_round(pid) do
    Agent.get_and_update(pid, fn state ->
      %{
        rounds_played: rounds_played,
        total_points: total_points,
        current_points: current_points
      } = state

      new_state = %{
        rounds_played: rounds_played + 1,
        total_points: total_points + current_points,
        current_points: 0
      }

      {new_state, new_state}
    end)
  end
end

{:ok, pid} = RunElixir.GameState.start_link([])
RunElixir.GameState.get_state(pid) |> IO.inspect(label: 1)
RunElixir.GameState.add_points(pid, 100) |> IO.inspect(label: 2)
RunElixir.GameState.get_state(pid) |> IO.inspect(label: 3)
RunElixir.GameState.end_round(pid) |> IO.inspect(label: 4)
1: %{current_points: 0, total_points: 0, rounds_played: 0}
2: :ok
3: %{current_points: 100, total_points: 0, rounds_played: 0}
4: %{current_points: 0, total_points: 100, rounds_played: 1}


  

    
ets
    

Erlang's ets module is the go-to solution for storing state that many processes need to read or write to. It can be customized for the performance requirements of your use case and allows performant concurrent read and write operations, unlike GenServers or Agents, which handle these requests sequentially and can quickly become the bottleneck of your system (although they can also handle many thousands of requests per second without breaking a sweat).
State in :ets is organized around dynamic tables, which means you can create and delete tables at any time without the need for migrations.
# Create a new :ets table with these options:
# :set - Create a Set table with unique key-value pairs
# :public - Allow all processes to access the table
# :named_table - Register the table under its name,
#                which allows processes to access it through
#                the atom name instead of the table PID.
:ets.new(:players, [:set, :public, :named_table])

# Insert values with the {key, value} format.
:ets.insert(:players, {:messi, %{age: 37, goals: 837}})
:ets.insert(:players, {:ronaldo, %{age: 39, goals: 900}})

# Look up a value
:ets.lookup(:players, :messi) |> IO.inspect(label: 1)

# Update a value by overwriting it
:ets.insert(:players, {:messi, %{age: 37, goals: 838}})
:ets.lookup(:players, :messi) |> IO.inspect(label: 2)

# Delete a record
:ets.delete(:players, :messi)
:ets.lookup(:players, :messi) |> IO.inspect(label: 3)
1: [messi: %{age: 37, goals: 837}]
2: [messi: %{age: 37, goals: 838}]
3: []

  
    
  
  Avoid memory bloat


Each table is one process and lives on until you terminate the table "owner" - the process that created it - or shut down the application. That means that unused tables aren't garbage collected and can bloat your memory usage if you don't clean them up properly.
So, when you stop using an ets table, you should delete the table with :ets.delete(:players). Alternatively, you can terminate the owner process and that will delete the table automatically.
If you want to terminate the table process without loosing the table, you can change the "owner" of the table with give_away/3 or by passing the PID of another process as the heir-option when you create the table.
defmodule RunElixir.TableOwner do
  @moduledoc "This is an example process meant to own an ets table."
  use GenServer
  require Logger

  def start(args), do: GenServer.start(__MODULE__, args)
  def init([name: name]), do: {:ok, name}

  # The process receives this message when it inherits the table.
  def handle_info({:"ETS-TRANSFER", table_name, from_pid, _data}, name) do
    Logger.info("#{name} inherited table #{table_name} from #{inspect(from_pid)}")
    {:noreply, name}
  end
end

{:ok, parent} = RunElixir.TableOwner.start([name: "Parent"])
{:ok, child} = RunElixir.TableOwner.start([name: "Child"])

IO.inspect(self(), label: "Livebook")
IO.inspect(parent, label: "Parent")
IO.inspect(child, label: "Child")

table_name = :inheritance

# Create a table with the Livebook process as parent and define the child
:ets.new(table_name, [:named_table, {:heir, child, nil}])
:ets.info(table_name) |> IO.inspect(label: "At Start")

# Give away the table from the current process to the Parent.
:ets.give_away(table_name, parent, nil)
:ets.info(table_name) |> IO.inspect(label: "Given away")

# Terminate the Parent, inherit the table to the Child
Process.exit(parent, :normal)
:timer.sleep(1) # <- Wait for the inheritance to happen
:ets.info(table_name) |> IO.inspect(label: "After inheritance")
Livebook: #PID<0.1582.0>
Parent: #PID<0.1669.0>
Child: #PID<0.1670.0>

At Start: [ owner: #PID<0.1582.0>, heir: #PID<0.1670.0> ]

Given away: [ owner: #PID<0.1669.0>, heir: #PID<0.1670.0> ]
16:47:10.686 [info] Parent inherited table inheritance from #PID<0.1582.0>

After inheritance: [ owner: #PID<0.1670.0>, heir: #PID<0.1670.0> ]
16:47:10.686 [info] Child inherited table inheritance from #PID<0.1669.0>

  
    
  
  Concurrency Options (ref)


You can optimize an ets table for concurrent write and read operations at the expense of increased memory consumption. All write operations stay atomic and isolated though. Simply add one of these options to the :ets.new/2 call:
	write_concurrency: false|true|auto	false - The default. Writes to the table must acquire a table-wide lock and block other writes until they finish.
	true - Writes to different records can happen in parallel, but writes to the same record still block each other.
	auto - Recommended over true for Erlang 25+. Similar to true, but it optimizes the synchronization granularity during runtime depending on how the table is used.


	read_concurrency: false|true	false - The default. Reads to the table must acquire a table-wide lock and block other reads until they finish.
	true - Optimizes the table for concurrent read operations, especially on machines with multiple CPUs. However, switching between read and write operations becomes more expensive.


	decentralized_counters: false|true	false - The default. Has no effect.
	true - Only has an effect for :ordered_set tables with write_concurrency: true. Optimizes the table for frequent calls that modify the table size (e.g. insert/2 and delete/2) at the expense of much slower calls to info/1.



When to use - and not to use - the concurrency flags:
	Enable read or write concurrency when you have many processes reading or writing to the table frequently, especially in bursts.
	nable both read and write concurrency, when you have frequent bursts coming from many processes for each, but not at the same time!.
	Don't enable read or write concurrency when only a few processes read or write to the table, if one operation occurs much more frequently than the other (e.g. many more reads than writes), or if you don't have bursts.
	Don't enable read and write concurrency when operations are interleaved (e.g. read/write/read/read/write/read).


  
    
  
  Table types


You can create tables with four different types (ref):
	:set - A Set table with unique, but unordered keys.
	:ordered_set - A Set table with unique, ordered keys.
	:bag - A Bag table with duplicate, unordered keys but multiple, unique values per key
	:duplicate_bag - A Bag table with duplicate, unordered keys and multiple, duplicate values per key

# :set tables keep only one key-value pair
set = :ets.new(:set, [:set])
:ets.insert(set, {1, :a})
:ets.insert(set, {1, :b})
:ets.insert(set, {1, :b})
:ets.lookup(set, 1) |> IO.inspect(label: 1)

# :ordered_set tables also keep only one key-value pair
ordered_set = :ets.new(:ordered_set, [:ordered_set])
:ets.insert(ordered_set, {1, :a})
:ets.insert(ordered_set, {1, :b})
:ets.insert(ordered_set, {1, :b})
:ets.lookup(ordered_set, 1) |> IO.inspect(label: 2)

# :bag tables keep multiple keys but only unique values
bag = :ets.new(:bag, [:bag])
:ets.insert(bag, {1, :a})
:ets.insert(bag, {1, :b})
:ets.insert(bag, {1, :b})
:ets.lookup(bag, 1) |> IO.inspect(label: 3)

# :duplicate_bag tables keep multiple keys and duplicate values
duplicate_bag = :ets.new(:duplicate_bag, [:duplicate_bag])
:ets.insert(duplicate_bag, {1, :a})
:ets.insert(duplicate_bag, {1, :b})
:ets.insert(duplicate_bag, {1, :b})
:ets.lookup(duplicate_bag, 1) |> IO.inspect(label: 4)
1: [{1, :b}]
2: [{1, :b}]
3: [{1, :a}, {1, :b}]
4: [{1, :a}, {1, :b}, {1, :b}]


  

    
dets
    

If you want a disk-based version of :ets, then dets is your solution, at least if your state fits into a file of less than 2 GB. Moving the state to disk could free up some memory, but it will come with a significant performance decrease because all operations are now disk operations.
# Define a path for the file. If the file doesn't exist, dets will create it.
# We have to convert the string to a charlist here, because that's what :dets expects.
dets_file_path = String.to_charlist("./state.dets")

# Open a new file
{:ok, _pid} = :dets.open_file(:state, file: dets_file_path)

# Write records to the file
:dets.insert(:state, {:messi, %{age: 37, goals: 837}})
:dets.insert(:state, {:ronaldo, %{age: 39, goals: 900}})

# Look up a value
:dets.lookup(:state, :messi) |> IO.inspect(label: 1)

# Update a value by overwriting it
:dets.insert(:state, {:messi, %{age: 37, goals: 838}})
:dets.lookup(:state, :messi) |> IO.inspect(label: 2)

# Delete a record
:dets.delete(:state, :messi)
:dets.lookup(:state, :messi) |> IO.inspect(label: 3)

# Flush the changes to disk. A file is auto-flushed every 3 minutes by default.
# You can change this interval with the `auto_save: 1_000` option in `:dets.open_file/2`
:dets.sync(:state)

# Close the file. Don't forget this, otherwise your file isn't closed properly
# and needs to be repaired the next time you open it, which might take a while.
#
# If you open the file from a process and the process terminates, this will be
# called automatically.
:dets.close(:state)
1: [messi: %{age: 37, goals: 837}]
2: [messi: %{age: 37, goals: 838}]
3: []


  

    
persistent_term
    

Erlang's persistent_term is a great option if you want to keep a large state in memory and you modify it rarely or ideally, never. It solves a problem that you'll run into if you use :ets or a GenServer to store a large state in memory: whenever a process fetches your large state, the BEAM will copy the entire state and return it to the calling process. Additionally, the state is copied again and again whenever the process uses it. If your state is large, this will seriously impact your processing speed and memory usage.
That's where persistent_term comes in. Instead of returning a copy of your state, it returns a reference to it. That way, only one version of the state stays in memory, but processes can still look up data and pass it around without problems.
# Let's download a "large file" (just 185kb but you get the point),
# parse it, and store its data in :persistent_term.
Mix.install([{:req, "~> 0.5"}, {:csv, "~> 3.2.1"}])

file = Req.get!("https://gist.githubusercontent.com/PJUllrich/05a091bb26669ba5fda57e5228885c95/raw/3ff9f16ae4679277e4f79cb044a860b0a0644892/medals.csv").body

# Convert the string body into a stream so that we can decode it.
{:ok, pid} = StringIO.open(file)

pid
|> IO.binstream(:line)
|> CSV.decode!(headers: true)
|> Enum.group_by(
  fn %{"country" => country, "discipline" => discipline} -> {country, discipline} end,
  fn %{"medal_code" => medal_code} -> medal_code end
)
|> Enum.each(fn {{country, discipline}, medals} ->
  :persistent_term.put({RunElixir.Medals, country, discipline}, medals)
end)

# Read the data from :persistent_term without copying it to the calling process.
#
# Because :persistent_term isn't table-based like ets or dets, it is common to
# prefix the keys with a module name that represents our data. In this case,
# we name our dataset "RunElixir.Medals".
:persistent_term.get({RunElixir.Medals, "Germany", "Judo"}) |> IO.inspect(label: 1)
:persistent_term.get({RunElixir.Medals, "Canada", "Swimming"}) |> IO.inspect(label: 2)
1: ["2"]
2: ["2", "1", "3", "1", "3", "2", "3", "1"]


  

    
Big-O Time Complexities
    


  
    
  
  Map


	Operation	Time Complexity
	Access	O(log n)
	Search	O(log n)
	Insertion	O(log n)
	Deletion	O(log n)
	Size	O(1)

Caveats:
	Maps are unordered, allow any key type, but disallow duplicate keys


  
    
  
  MapSet


Same complexities as 'Map'

  
    
  
  List


	Operation	Time Complexity
	Access	O(n)
	Search	O(n)
	Insertion	O(1) for prepending, otherwise O(n)
	Deletion	O(1) if first element, otherwise O(n)
	Length	O(n)

Caveats
	Lists are not arrays as in other languages, but single-linked lists


  
    
  
  Keyword (List)


A Keyword (list) has the same time complexities as List.
Every entry in a Keyword (list) is a tuple with the first element being the key and the second element the value.
keyword = [{:a, 1}, {:b, 2}]
# Can also be written as:
keyword = [a: 1, b: 2]
Caveats
	Keys must be atoms.
	Keys are ordered, as specified by the developer.
	A keyword list may have duplicate keys, but duplicates are often ignored by functions like Keyword.get/3 (returns the first match) and are even removed by e.g. Keyword.put/3 and Keyword.delete/2.iex> Keyword.get([{:a, 1}, {:a, 2}], :a)
1

iex> Keyword.put([{:a, 1}, {:a, 2}], :a, 3)
[a: 3]

iex> Keyword.delete([{:a, 1}, {:a, 2}], :a)
[]



  
    
  
  Tuple


	Operation	Time Complexity
	Access	O(1)
	Search	O(n)
	Insertion	O(n)
	Deletion	O(n)
	Length	O(1)

Caveats
	Tuples are better for reading, whereas Lists are better for traversals
	Avoid using tuples as a collection	(i.e. avoid Tuple.append/2, Tuple.insert_at/3, or Tuple.delete_at/2)




  
    
  
  (erlang) array


	Operation	Time Complexity
	Access	O(log n) [7]
	Search	O(log n)
	Insertion	O(log n)
	Deletion	O(log n)
	Length	O(log n)

Caveats
	An array is a trie of small tuples, with a smaller memory overhead to Lists
	Deletion is actually a replace with the default value and creates sparse arrays	For real deletion, use sparse_to_list/1, which has O(n) complexity




  
    
  
  :queue


	Operation	Time Complexity
	Access	O(1) for first and last element, O(n) otherwise
	Search	O(n)
	Insertion	O(1)
	Deletion	O(n)
	Length	O(n)


  
    
  
  :ets


	Operation	Time Complexity
	Access/Search/Insertion	O(1) for set, O(log n) for ordered_set where n is the table size, O(n) for bag and duplicate_bag where n is the number of duplicate keys
	Deletion	O(1) for set, O(n) for ordered_set (ref), ? for bag and ordered_bag
	Length	?, depends on the decentralized_counters table option (ref)


  
    
  
  :persistent_term


	Operation	Time Complexity
	Access	O(1)
	Search	O(1)
	Insertion	O(n) where n is the number of references to the term
	Deletion	Just don't (ref)
	Length	?


  
    
  
  References


	Partial overview of complexities
	Discussion of :array
	Does Elixir have persistent data structures?
	Way to get O(1) access/set
	Sequences by sasajuric



  

    
Basic Type Hierarchy
    

When you compare basic types in Elixir, they adhere to the following hierarchy. This is important to keep in mind when comparing values with < or > comparison operators.
This is the order of the basic types for comparisons:
	Number
	Atom
	Reference
	Function
	Port
	PID
	Tuple
	Map
	List
	String

There are two types on this list that we haven't discussed before: Reference and Port.
A Reference is a value that is unique across all connected erlang nodes. So, if you connect multiple nodes/servers into one cluster and call make_ref() on any of them, you will always get a unique value back. References are useful for tagging requests and to track them throughou your system. They also come in handy for tagging messages that you send between processes, especially when the response is asynchronous.
make_ref()
#Reference<0.2725402356.1941176323.86081>
A Port lets you communicate with the "external world", for example with the operating system of the server your Elixir program runs on. You can use it to call functions in other programming languages easily, like Zig or Rust.
Port.list()
[#Port<0.0>, #Port<0.1>, #Port<0.3>]
Let's see what that the ordering means in practice:
ref = make_ref()
port = Port.list() |> hd()
pid = self()

# A Number is "larger" than an Atom,
# which is "larger" than a Reference, and so on.
result = 1 < :atom < ref < port < pid < {:a, :b} < %{a: 1} < [1, 2] < "Elixir"

IO.inspect(result)
true

